Errore inerente



Errore di troncamento (Formula di Taylor)

```
s1=Normal[Series[Sin[x],{x,0,3}]]
```

$$x - \frac{x^3}{6}$$

```
s2=Normal[Series[Sin[x],{x,0,10}]]
```

$$x - \frac{x^3}{6} + \frac{x^5}{120} - \frac{x^7}{5040} + \frac{x^9}{362\,880}$$

```
Plot[Evaluate[{s1, s2, Sin[x]}], {x, 0, 2 π}, PlotStyle → {Green, Red, Black}]
```



Errore di arrotondamento

```
f[x_]:= Evaluate[Expand[(x-1) (x-1) (x-1) (x-2)]]
```

```
Plot[f[x], {x, 0.5, 1.5}, PlotStyle → Red]
```




```
ll[i_] := Table[{N[10^5 x], 10^15 f[N[x + 1]]}, {x, - i/10^5, i/10^5, 2 i/10^7}];
```

```
GraphicsGrid[
 Partition[
  Table[
   ListPlot[ll[i], Joined → True,
    PlotStyle → {Thickness[0.005], Red}], {i, 4, 1, -1}], 2]]
```



Tutto insieme



Catastrofi

cancellazione

```
Clear[b,f,f1]
```

```
f[b_]:=1-Sqrt[1+b^2]
```

```
f[10^-8]
```

$$1 - \frac{\sqrt{10\,000\,000\,000\,000\,001}}{100\,000\,000}$$

```
f[10.^-8]
```

0.

```
N[f[10.^-8],30]
```

0.

```
N[f[10^-8],30]
```

$-4.99999999999999998750000000000 \times 10^{-17}$

```
N[Sqrt[1+b^2],40]/.b->10^-8
```

1.000000000000000049999999999999999875000000

```
Series[1-Sqrt[1+b^2],{b,0,2}]
```

$$-\frac{b^2}{2} + O[b]^3$$

```
f1[b_]:=-b^2/2
```

```
f1[10.^-8]
```

$-5. \times 10^{-17}$

Instabilità

```
Clear[f];
f[x_]:=4. x(1-x)
```

```
n=200;
```

```
lp[z_, c_] := ListPlot[NestList[f, 0.1` + N[z], n], Joined → True, PlotStyle → c];
```

$$\text{Show}\left[\text{lp}[0, \text{Red}], \text{lp}\left[\frac{1}{10^{10}}, \text{Blue}\right], \text{AspectRatio} \to \frac{1}{3}\right]$$



Serie a termini alterni

```
ss = Series[Exp[x], {x, 0, 10}]
```

$$1 + x + \frac{x^2}{2} + \frac{x^3}{6} + \frac{x^4}{24} + \frac{x^5}{120} + \frac{x^6}{720} + \frac{x^7}{5040} + \frac{x^8}{40\,320} + \frac{x^9}{362\,880} + \frac{x^{10}}{3\,628\,800} + O[x]^{11}$$

```
sn = ss // Normal
```

$$1 + x + \frac{x^2}{2} + \frac{x^3}{6} + \frac{x^4}{24} + \frac{x^5}{120} + \frac{x^6}{720} + \frac{x^7}{5040} + \frac{x^8}{40\,320} + \frac{x^9}{362\,880} + \frac{x^{10}}{3\,628\,800}$$

```
sn = Normal[Series[Exp[x], {x, 0, 10000}]];
```

calcolo per x = - 10

```
sn /. x → -10.
```

0.0000453999

```
1 / (sn /. x → 10.)
```

0.0000453999

```
N[Exp[-10]]
```

0.0000453999

calcolo per x = - 100

```
sn /. x → -100.
```

5.14084×10^{26}

```
1 / (sn /. x → 100.)
```

3.72008×10^{-44}

```
N[Exp[-100]]
```

3.72008×10^{-44}

Calcolo della derivata

f è una funzione

```
f=Cos[x]
```

Cos[x]

f1 la sua derivata

```
f1=D[f,x]
```

-Sin[x]

f1app1 la sua approssimazione con il rapporto incrementale

```
f1app1[h_]:=(Cos[x+h]-Cos[x])/h
```

L10 definisce la funzione -Log10 |x| con una protezione ...

```
L10[x_]:=If[Abs[x]<10^-20,20,N[-Log[10,Abs[x]]]];
```

```
err[k_]:=L10[(f1-f1app1[10^-k])/.x->1.1]
```

```
Plot[err[h], {h, 0, 15}, PlotStyle → {Red}]
```



```
Series[f1app1[h]-f1,{h,0,3}]
```

$$-\frac{1}{2}\text{Cos}[x]\,h+\frac{1}{6}\text{Sin}[x]\,h^2+\frac{1}{24}\text{Cos}[x]\,h^3+O[h]^4$$

```
(f1app1[h]+f1app1[-h])/2
```

$$\frac{1}{2}\left(-\frac{\text{Cos}[h-x]-\text{Cos}[x]}{h}+\frac{-\text{Cos}[x]+\text{Cos}[h+x]}{h}\right)$$

```
Together[%]
```

$$\frac{-\text{Cos}[h-x]+\text{Cos}[h+x]}{2\,h}$$

```
f1app2[h_]:=(Cos[x+h]-Cos[x-h])/(2h)
```

```
Series[f1app2[h]-f1,{h,0,3}]
```

$$\frac{1}{6}\text{Sin}[x]\,h^2+O[h]^4$$

```
err2[h_]:=L10[(f1-f1app2[10^-h])/.x->1.1]
```

```
Plot[{err[h], err2[h]}, {h, 0, 15}, PlotStyle → {Red, Blue}]
```




```
Series[f1app2[h]-f1,{h,0,5}]
```

$$\frac{1}{6}\text{Sin}[x]\,h^2-\frac{1}{120}\text{Sin}[x]\,h^4+O[h]^6$$

```
Series[f1app2[h/2]-f1,{h,0,5}]
```

$$\frac{1}{24}\text{Sin}[x]\,h^2-\frac{\text{Sin}[x]\,h^4}{1920}+O[h]^6$$

```
Solve[{a+b==1,a/6+b/24==0}]
```

{}

```
-1/3 f1app2[h] + 4/3 f1app2[h/2]
```

$$\frac{4\left(-\text{Cos}\left[\frac{h}{2}-x\right]+\text{Cos}\left[\frac{h}{2}+x\right]\right)}{3\,h}-\frac{-\text{Cos}[h-x]+\text{Cos}[h+x]}{6\,h}$$

```
f1app3[h_]:= -1/3 f1app2[h] + 4/3 f1app2[h/2]
```

```
Series[f1app3[h]-f1,{h,0,5}]
```

$$\frac{1}{480}\,\mathrm{Sin}[x]\,h^4+O[h]^6$$

```
err3[h_]:=L10[(f1-f1app3[10^-h])/.x->1.1]
```

```
Plot[{err[h], err2[h], err3[h]}, {h, 0, 15}, PlotStyle → {Red, Blue, Green}]
```



Calcolo di π con un metodo Montecarlo

1000 punti



10000 punti



100000 punti



programma semplificato

```
rand:=Random[Real,{-1,1}];
```

```
i=0;
j=1000000;
Do[If[rand^2+rand^2 < 1,i++],{j}];
N[4 i/j]
3.14248
```

Con un generatore cattivo